

## Paper 214-2011


## Annotated Plot of z/OS Auxiliary Storage Local Page Slot Allocation Using SAS/GRAPH® and MXG® software

Neal Musitano Jr.

U. S. Department of Veterans Affairs, Information Technology Center  
Philadelphia, Pennsylvania

### ABSTRACT

This paper is about creating an annotated SAS/GRAPH plot of z/OS® local page dataset slot utilization. The z/OS local page dataset slots are used by z/OS for management of system memory. The annotated plot shows the count of the local page slots defined and the number of slots that are used. A red star highlights the maximum local slot usage. Through the day, slots are used and released with copies of main memory information. For z/OS performance reasons the MXG slot utilization guideline is at or below 25%.



At higher utilizations, z/OS response times can degrade. And an extreme local page slot shortage could even cause the z/OS server to crash. The data for the SAS/GRAPH plot is from the z/OS Resource Measurement Facility record type 71. The MXG performance database (PDB) stores this record information in SAS format in the PDB.TYPE71 dataset.

By plotting the daily local page slot usage our z/OS system support personnel are able easily verify that no shortage exists or to research potential problems that could contribute to a shortage. Such problems could be, insufficient central storage, or not allocating enough local page datasets with sufficient slots, or perhaps an unusual condition that never released unused slots. Each problem condition may require a different technical response to resolve the problem and to prevent a system outage. SAS and SAS/GRAPH help in daily z/OS monitoring to prevent any z/OS outage from a lack of local page slots.

This paper specifically details annotating z/OS RMF data in the MXG PDB.TYPE71 dataset and uses a user macro to plot the TYPE71 data variables. However, the annotated GPLOT example can be adapted to produce a variety of daily monitoring plots by using different input files and variables.

### KEYWORDS – z/OS MXG SAS/GRAPH ANNOTATE GPLOT

### INTRODUCTION

On the z/OS platform the RMF record type seventy-one records are produced as part of the daily System Management Facility (SMF) processing. MXG software processes this record type and many others into a Performance Database (PDB) consisting of a hundred or more MXG/SAS datasets. The MXG PDB TYEP71 dataset contains detailed z/OS paging activity, with each observation containing the record sorted by SYSPLEX, SYSTEM, SYSNAME and timestamp.

The z/OS SMF/RMF record types vary from record zero to record type two-hundred-fifty-five and provide additional detail for z/OS performance monitoring. The SMF/RMF record types are used to build numerous MXG SAS datasets on z/OS system activity. Each MXG PDB dataset contains performance variables in a SAS readable format that can be processed by SAS tasks. MXG also provides numerous examples for analysis and plots using BASE SAS and SAS/GRAPH.

## Z/OS JOB CONTROL LANGUAGE (JCL) SETUP & PREPARATION

The daily z/OS MXG processing builds the MXG Performance Database including the PDB.TYEP71 dataset used in this example. The MXG PDB is the z/OS dataset HLQ.MXG.PDB.DAILY (0). HLQ is an abbreviation for high-level qualifier, the first qualifier of the z/OS dataset. Note, that many SAS datasets can exist within the one z/OS MXG Performance Database dataset.

The daily MXG PDB can be processed using the MXGSASV9 JCL procedure or with the standard SAS procedure that is shipped with SAS version 9 and used in this example.

The Figure #1 JCL example below uses LIBRARY and SOURCLIB DD cards to reference MXG FORMATS and SOURCLIB datasets, these datasets are used for most MXG processing and reports. The MXG FORMATS dataset is created with the installation of the MXG software and includes numerous formats for the MXG variables.

The PDB DD card is used to reference the MXG/SAS TYPE71 data member within the PDB.

### Figure #1 SAS BATCH JOB JCL TO READ THE DAILY MXG PDB

```

//*****
//* DOC: JCL FOR MXG PDB TYPE71 SAS GRAPHS *
//*****
//SGF2011 EXEC SASV9,
//  OPTIONS 'MACRO SYSPARM=SGFPATH'
//LIBRARY DD DSN HLQ.MXG.FORMATS,DISP SHR
//SOURCLIB DD DSN HLQ.MXG.USER.SOURCLIB,DISP SHR
//          DD DSN HLQ.MXG.SOURCLIB,DISP SHR
//PDB      DD DSN HLQ.MXG.PDB.DAILY(0),DISP SHR
//SYSPRINT DD SYSOUT=*
//SYSOUT   DD SYSOUT=*
//SYSIN    DD *
/** SLOTTUSED - PLOT DAILY LOCAL PAGE SLOT USAGE INFORMATION **/

```

Since the program will process my user macros the MACRO option is used to allow SAS to recognize the macro language statements, calls, and variable references. The SYSPARM SAS option is used to specify a character string that is passed from z/OS to the SAS program. This example string of 'SGFPATH' is available to the program via the SAS &SYSPARM automatic macro variable and is used to select the application path for the graph.

The JCL was edited using the z/OS Time Sharing Option, Interactive System Productivity Facility TSO/ISPF or simply TSO, and the "HILITE ON" setting was used to make the JCL easier to read. From the edit session the PF1 key can be used for HELP or the PROFILE command can be used to check the edit session setting.

## WHAT'S IN THE DAILY MXG PERFORMANCE DATABASE

SAS PROC CONTENTS is used to review the details of what is in the daily PDB. The SAS code in Figure #2 provides an example to list the contents of all the datasets in the daily PDB and the directory. Sometimes it's a good idea to print or direct the output to a file for reference, since the contents report of the daily MXG PDB can exceed 15,000 lines out output.

The output in Figure #3 and Figure #4 are partial listing of the z/OS PDB dataset directory with its SAS datasets and more details on the hundreds of variables in the PDB.TYPE71 file of the PDB. The directory option lists over 100 files (i.e. JOBS, STEPS, TAPES, TYPE70, TYPE71, etc.) in the PDB. And given the number of variables in each file, the output listing becomes too large to include within this paper.

**Figure #2 PROC CONTENTS TO DISPLAY PDB INFORMATION**

```
PROC CONTENTS DATA=PDB._ALL_ DIRECTORY ;
```

**Figure #3 PROC CONTENTS DIRECTORY OUTPUT AND LIST OF MEMBERS**

```

                                THE SAS SYSTEM
                                THE CONTENTS PROCEDURE
                                DIRECTORY
LIBREF                          PDB
ENGINE                          BASE
ACCESS                          READONLY
PHYSICAL NAME                   HLQ.MXG.PDB.DAILY.G1234V00
UNIT                             DISK
VOLUME                          CAP001
DISPOSITION                     SHR
...more info on DEVICE , BLOCKSIZE, BLOCKS PER TRACK , TOTAL BLOCKS, etc
MEMBERS                          104

                                NUMBER
                                OF
#  NAME      MEMBER  TYPE  PAGESIZE  PAGES  CREATED          LAST MODIFI
1  ASUMCEC   DATA    46080    10   10MAR10:00:26:00  10MAR10:
2  ASUMCELP DATA    23040     3   10MAR10:00:25:59  10MAR10:
..  more members.

58 TYPE70    DATA    46080    14   10MAR10:00:11:06  10MAR10:
...more members

63 TYPE71    DATA    46080     5   10MAR10:00:11:15  10MAR10:

list continues for 100+ member datasets

```

**Figure #4 SNIP OF OUTPUT FROM THE CONTENTS PROCEDURE**

```

DATA SET NAME      PDB.TYPE71      OBSERVATIONS      96
MEMBER TYPE       DATA      VARIABLES          400
ENGINE            BASE      INDEXES             0
CREATED           MONDAY, MAY 10, 2010 12:05:34 AM  OBSERVATION LENGTH 2003
LAST MODIFIED     MONDAY, MAY 10, 2010 12:05:34 AM  DELETED OBSERVATIONS 0
PROTECTION
DATA SET TYPE
LABEL             TY71: RMF PAGING ACTIVITY  COMPRESSED        CHAR
DATA REPRESENTATION MVS_32      REUSE SPACE       NO
ENCODING          EBCDIC1047 WESTERN (EBCDIC)  POINT TO OBSERVATIONS YES
                                                         SORTED            YES

```

ENGINE/HOST DEPENDENT INFORMATION

```

DATA SET PAGE SIZE      46080
NUMBER OF DATA SET PAGES 5
NUMBER OF DATA SET REPAIRS 0
PHYSICAL NAME          HLQ.MXG.PDB.DAILY.G5678V00
RELEASE CREATED        9.0101M3
RELEASE LAST MODIFIED  9.0101M3
CREATED BY             NSGFMXGB
LAST MODIFIED BY      NSGFMXGB
SUBEXTENTS             1
TOTAL BLOCKS USED     10

```

ALPHABETIC LIST OF VARIABLES AND ATTRIBUTES

```

..
# VARIABLE  TYPE  LEN  FORMAT  LABEL
26 ASMNVSC  NUM   5          LOCAL SLOTS*ALLOCATED*NON-VIO
27 ASMSLOTS NUM   5          LOCAL*SLOTS*DEFINED
28 ASMSLOTX NUM   5          LOCAL*SLOTS*UNALLOCATED
29 ASMVSC   NUM   5          LOCAL SLOTS*ALLOCATED*VIO
30 AVLEXTAV NUM   5          AVAILABLE*EXT STORAGE*AVG PAGES
31 AVLEXTMN NUM   5          AVAILABLE*EXT STORAGE*MIN PAGES
32 AVLEXTMX NUM   5          AVAILABLE*EXT STORAGE*MAX PAGES
33 BLKPAGE  NUM   5  5.1      BLOCKED*PAGES*PAGED IN*PER SEC
 7 BLKSAUIN NUM   5          BLOCKS*PAGED IN/SEC*FROM AUX*PER SEC
34 COMPAGE  NUM   5  5.1      TOTAL*COMMON*PAGING RATE
35 CPUPAGTM NUM   5  TIME12.2 CPU TIME*FOR PAGE MOVEMENT*IN CSTORE
..
..hundreds of additional variables
..
3 ZDATE    NUM   4  DATE9.    ZEE DATE*ZEE OBS*WAS CREATED

```

SORT INFORMATION

```

SORTEDBY      SYSPLEX SYSTEM SYSNAME STARTIME SMFTIME
VALIDATED     YES
CHARACTER SET EBCDIC
SORT OPTION    NODUPREC

```

## SAS PROGRAM PREPARATION

The program code is displayed on the next page in Figure #5. Since the SAS/GRAPH plot will be produced using PROC GPLOT with “BY” variables and “BY” values placed in the titles, the NOBYLINE option is used to keep the default byline from displaying in the graph. This makes for a more professional looking graph.

SAS program maintenance for the program is simplified by the use of &LET statements. This is done to identify the input dataset. Referencing the TYPE71 dataset via the %LET statement, minimizes possible errors in referencing the dataset through the program, and the dataset name can be displayed in the footnotes of the graphics output without error. This is a great feature and makes the program code useful for charting several other datasets and variables.

The path to store the graph is defined at the beginning of the program by using a user macro named SET\_USER\_PATH. An input parameter from the JCL is used via the &SYSPARM automatic macro variable to select the user path desired, i.e. for test graphs or SGF graphs as needed. The macro variable &USERPATH is then created with the actual path and made available to the program. It is up to the user to make sure the paths defined actually exist. This example defines a path to a UNIX type directory on the z/OS system with the day of the month concatenated to the daily path.

The example reads the PDB.TYPE71 data and creates a working TYPE71 dataset i.e. WORK.TYPE71. The “WORK” part of the dataset name is a default and doesn’t have to be specified in the references that follow. This dataset contains information on the TYPE71 observations, plus the time and date variables extracted from the STARTIME variable. While it is technically possible to plot the data using the STARTIME variable, the charts are easier to read and more acceptable to most managers, if the date is in the title and time is used for the horizontal axis.

MXG calculates the slot utilization variable of SLOTUTIL using the SLOTL0MN and SLOTL0NMN variables. Thus this example uses the SLOTL0MN and SLOTL0NMN variables to calculate a new variable labeled “slots used up” to view the local page slot usage. The plot variables are input to the USER\_TYPE71\_PLOT and plotted on an overlay graph. The peak SLOTUTIL and time of the peak is also displayed in the annotated plot.

A threshold of twenty-five percent of SLOTL0MN is calculated and displayed in the graph. The 25% slot usage is a technical MXG recommendation or guideline, reference MXG change 15.064 in the MXG newsletters or member CHANGE55. Since page slot usage of above that threshold starts to degrade the paging processing time and slows most everything on the z/OS system. Depending on the situation, more page datasets or larger page datasets might be needed to lower the utilization.

The user plotting macro USER\_TYPE71\_PLOT for the TYPE71 data is in the SGFMACS member in the user sourclib dataset as is the macro SET\_USER\_PATH. The macro’s are made available to the SAS program via the sourclib statement in the JCL and the %INCLUDE statement in the SAS program. The first plot variable PLOTVAR1 of the macro is used for the red star “y” axis annotation. The source code for both macros is available in the appendix.

After the SAS/GRAPH plot on local page slot utilization is produced, the macro can be used again to produce additional plots using different variables from the TYPE71 data.

Figure #5 SAS CODE - IDENTIFY THE INPUT FILE AND VARIABLES TO PLOT

```

/** SLOUSED - PLOT DAILY LOCAL PAGE SLOT USAGE INFORMATION **/
OPTIONS NODATE NONUMBER NOBYLINE;          /** SAS OPTIONS          **/
%INCLUDE SOURCLIB(SGFMACS) ;                /** INCLUDE MY MACROS      **/

/** IDENTIFY THE MXG PDB DATASET TO PROCESS AND VAR(S) **/
%LET MY_FILE = TYPE71;
%LET THRESHOLD = .25;      /** USED UP LOCAL SLOT THRESHOLD **/
%LET MY_NOTE=
      H=2.3 PCT C=RED FONT=SWISSB
      MOVE=(58,71) "AT 25% THRESHOLD - REVIEW USAGE";

/** SET THE USERPATH FOR THE PROGRAM GRAPHICS **/
%SET_USER_PATH

/** PROCESS THE PDB FILE BEFORE MOST CHARTS OR PLOTS **/
DATA &MY_FILE;
LABEL TIME          = 'TIME*OF*DAY'
      SLOUSED      = 'LOCAL*SLOTS*USED*UP' ;
SET PDB.&MY_FILE;
TIME = TIMEPART(STARTIME);
DATE = DATEPART(STARTIME);
SLOUSED = SLOTL0MN - SLOTL0NMN;  /** LOCAL SLOTS USED **/
THRESHOLD = SLOTL0MN * &THRESHOLD; /** 25% THRESHOLD LINE**/
RUN;

/** NEXT USE A MACRO TO ANOTATE AND PLOT DESIRED VARIABLES **/
%USER_TYPE71_PLOT ( MXGFILE=&MY_FILE,
                  PLOTVAR1=SLOUSED,
                  PLOTVAR2=SLOTL0MN,
                  PLOTVAR3=THRESHOLD,
                  ANNO_VAR2=SLOTL0M,
                  TY71NOTE=&MY_NOTE,
                  TY71PATH=&USERPATH          /** NO COMMA AFTER **/
                  )                          /** LAST ARGUMENT **/


/** PRODUCE ANOTHER TYPE71 PLOT USING DIFFERENT VARIABLES **/
%USER_TYPE71_PLOT ( MXGFILE=&MY_FILE,
                  PLOTVAR1=HIUICMX,
                  PLOTVAR2=HIUICMN,
                  PLOTVAR3=HIUICAV,
                  TY71PATH=&USERPATH
                  )
RUN;

```

## REVIEW THE ANNOTATED SAS/GRAPH GLOT OUTPUT

The annotation will place a red star at the maximum slot utilization, i.e. the maximum slots used up value, i.e. &PLOTVAR1 value. The PROC SORT by MVSLEVEL, SYSPLEX, SYSTEM, SYSNAME and DATE places the maximum &PLOTVAR1 as the first observation of the “BY” variables in the sorted annotation dataset. Thus, if the input data includes multiple systems, dates, etc, they will be sorted and the maximum &PLOTVAR1 will be the first observation in the annotation dataset for each.

**Figure #6 ANNOTATED SAS/GRAPH PROC GLOT OUTPUT**



### Annotated Plot of z/OS® Auxiliary Storage Local Page Slot Allocation

Using simulated data, the above graph displays the local page slot usage and highlights the time of the peak usage. The Z/OS version is displayed in the title along with the sysplex name, system id, and system name, and finally the date in weekdate29 format. All the above information is taken from the data in the PDB.TYPE71 file, eliminating the need for special edits or code to update the graph for special management reports.

The peak slot utilization in the graph occurred at 9:30 am with a value of 37%. That is above the 25% recommended threshold, and some additional local page slots could be used. The footnotes contain the MXGFILE i.e. &MY\_FILE and &PLOTVAR1. This helps to document the graphic plot specifics, the input MXG database file and to minimize any confusion about what the graph is about.

## CONCLUSION

The z/OS Base SAS JCL procedure is used in this example, and a daily job is scheduled by automated software to produce the daily annotated plot of z/OS® auxiliary storage local page slot usage. Using SAS and MXG software for monitoring auxiliary storage usage that approaches known set thresholds is an efficient use of our computer resources, software and personnel

Producing SAS/GRAPH output via SAS Output Delivery System (ODS) in GIF format for viewing via an intranet Web browser is also appealing to many staff members and management. The display of graphs on the intranet web server improved ease of use in reviewing performance data and increased the use of MXG and SAS/GRAPH software. Overall productivity increased by making the charts and reports more easily available for review by the staff and managers needing access to the information

## REFERENCES

IBM z/OS - MVS System Management Facilities (SMF) – [SA22-7630-17](#)  
IBM z/OS – Resource Measurement Facility (RMF) User's Guide – [SC33-7990-11](#)  
IBM Redbooks at <http://www.redbooks.ibm.com/redbooks.nsf/redbooks/>  
Merrill's Expanded Guide to Computer Performance Evaluation Using the SAS System  
Merrill's Expanded Guide Supplement and Technical Newsletters for Users of MXG®  
MXG Archives at [WWW.MXG.COM](http://WWW.MXG.COM)  
Computer Measurement Group Information at <http://www.cmg.org/>  
SHARE Information at <http://www.share.org>  
Philadelphia SAS Users Group Presenter Information at <http://www.philasug.org/call4pap.htm>  
NorthEast SAS Users Group information and proceedings at <http://www.nesug.org>  
SAS Technical Support at <http://support.sas.com>  
SAS-L archives at <http://www.listserv.uga.edu>  
SAS Online Documentation at <http://support.sas.com/documentation/onlinedoc>  
SAS Proceedings at <http://support.sas.com/events/sasglobalforum/previous/online.html>  
SAS Global Forum Guidelines <http://support.sas.com/events/sasglobalforum/2011/package.html>

## CONTACT INFORMATION

Your comments and questions are welcome.

Neal Musitano Jr.  
Department of Veterans Affairs  
Information Technology Center-284/31  
P. O. BOX 7545  
Philadelphia, PA. 19101-7545

Phone 215-842-2000 ext. 4102    EMAIL [Neal.Musitano@va.gov](mailto:Neal.Musitano@va.gov)



SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.

---

Note: A modified test MXG/PDB was used for the information displayed, not to include the actual values of detailed production computer performance data in the paper.

## APPENDIX – MACRO CODE.

Text can be copied from the PDF document by using the ‘select text tool’ icon on the PDF toolbar, or by a mouse right click and selecting the select tool.

```

%MACRO SET_USER_PATH                /** MACRO TO SET THE USERPATH    **/
%GLOBAL USERPATH;                  /** USERPATH AVAILABLE TO PROGRAM **/
DATA _NULL_;                        /** DEFINE ODS GRAPHIC USERPATH **/
  PATH = "&SYSPARM" ;                /** INPUT FROM JCL OPTION    **/
  IF PATH = 'SGFPATH' THEN
    CALL SYMPUT('USERPATH', '/u/sgf/' || PUT(DAY(TODAY()-1), Z2.0));
  ELSE IF PATH = 'TESTPATH' THEN
    CALL SYMPUT('USERPATH', '/u/tst/' || PUT(DAY(TODAY()-1), Z2.0));
  ELSE CALL SYMPUT('USERPATH', '/nopath/');
RUN;
%MEND SET_USER_PATH;                /** END OF SET USER PATH MACRO **/

%MACRO USER_TYPE71_PLOT
(MXGFILE = ,      PLOTVAR1=,      PLOTVAR2=,      PLOTVAR3=,
 ANNO_VAR2=,     TY71NOTE=,     TY71PATH="/u/mxg/01/" );

/** SORT TO FIND ANNOTATION VARIABLE VALUE FOR ANNOTATION **/
PROC SORT DATA=&MXGFILE  OUT=MY_ANNOTATE;
  BY MVSLEVEL SYSPLEX SYSTEM SYSNAME DATE DESCENDING &PLOTVAR1;
RUN;
/** SELECT FIRST BY VALUE OBSERVATION(S) ANNOTATE MAXIMUM **/
/** TIME OF MAXIMUM AND INFORMATION FOR ANNOTATION **/
DATA MY_ANNOTATE ;
  LENGTH FUNCTION STYLE COLOR $ 8 TEXT $27;
  RETAIN XSYS '2' YSYS '2' HSYS '1' WHEN 'a' ;
  SET MY_ANNOTATE ;      BY MVSLEVEL SYSPLEX SYSTEM SYSNAME DATE;
  IF FIRST.MVSLEVEL OR FIRST.SYSPLEX OR FIRST.SYSTEM
  OR FIRST.SYSNAME OR FIRST.DATE;
  FUNCTION='LABEL';      /** ANNOTATE FUNCTION    **/
  COLOR='RED';           /** ANNOTATE STAR COLOR  **/
  STYLE='SPECIAL'; TEXT='M'; /** STYLE SELECT FOR STAR**/
  SIZE=10;              /** SIZE INFORMATION    **/
  X=TIME; Y=&PLOTVAR1;  /** ANNO X,Y LOCATION  **/
  POSITION='B'; OUTPUT;  /** CENTER ABOVE X, Y  **/

  COLOR='PURPLE';       /** ANNOTATE TEXT COLOR **/
  SIZE=6; ANGLE=1 ;    /** ANNOTATE TEXT INFO  **/
  STYLE='SWISSBI';
  TEXT= ' ' || left(put (&PLOTVAR1, COMMA10.0))
        || ' at ' || left(put(time,time5.0))
  POSITION='3'; OUTPUT ; /** LEFT CELL ABOVE X, Y **/

  /* MACRO FILTER – IF ANNOTATE VARIABLE2 NOT SPECIFIED */
  %IF %LENGTH(&ANNO_VAR2) > 0 %THEN %DO;
  POSITION='6';          /** LEFT CELL ABOVE X, Y **/
  TEXT= ' ' || left(put (&ANNO_VAR2, COMMA4.0)) || ' %';
  OUTPUT;
  %END;
RUN;

ODS LISTING CLOSE;                /** CLOSE LISTING    **/
ODS HTML PATH="&TY71PATH" (URL=NONE) /** WEB PATH INFO    **/
  BODY="DAILY_&MXGFILE..&PLOTVAR1..HTML"
  (TITLE="SAS/MXG PLOT &MXGFILE &PLOTVAR1");

```

```

GOPTIONS RESET=GLOBAL          /** RESET GRAPHIC OPTIONS**/
GUNIT=PCT  DEVICE=GIF          /** GRAPH UNITS PERCENT  **/
XMAX=10.0IN YMAX=6.0IN        /** GRAPH X Y SIZE      **/
CBACK=CXFFFFCC                /** GRAPH BACKGRND COLOR **/
FTEXT=SIMPLEX CTEXT=BLUE HTEXT=2.2 ;

PROC GPLOT DATA=&MXGFILE ; BY MVSLEVEL SYSPLEX SYSTEM SYSNAME DATE;
FORMAT DATE          WEEKDATE29.
      TIME           TIME5.0
      &PLOTVAR1     COMMA9.0;

TITLE1 HEIGHT=3.0 C=BLUE 'Z/OS' C=RED '#BYVAL1 '
      C=BLUE 'COMPUTER SYSTEM PLOT OF'
      C=RED " &PLOTVAR1";
TITLE2 HEIGHT=3.0 C=BLUE '#BYVAR2= ' C=RED '#BYVAL2 '
      C=BLUE ' #BYVAR3= ' C=RED '#BYVAL3 '
      C=BLUE ' #BYVAR4= ' C=RED '#BYVAL4';
TITLE3 HEIGHT=2.8 C=VIRO '#BYVAL5';

AXIS1 WIDTH=2 COLOR=BLUE          /** VERT RESPONSE AXIS  **/
      LABEL=(H=2.2 COLOR=CYAN ROTATE=90 ANGLE=-90)
      MAJOR=(H=2 C=GREEN W=2)      /** AXIS MAJOR TICK MARK **/
      MINOR=NONE                   /** MINOR TICK MARKS   **/
      VALUE=(H=2.8 COLOR=CYAN);    /** RESPONSE VALUE INFO **/
AXIS2 WIDTH=2                    /** HORIZONTAL TIME AXIS **/
      LABEL=(HEIGHT=2.9 C=RED)     /** TIME*OF*DAY LABEL   **/
      MAJOR=(HEIGHT=2 C=RED)       /** AXIS MAJOR TICK MARKS**/
      MINOR=(H=1 N=5 C=BIB)       /** AXIS MINOR TICK MARKS**/
      ORDER=('00:00'T TO '24:00'T BY '06:00'T)
      VALUE=(H=2.9 C=GREEN);       /** TIME*OF*DAY VALUES **/
FOOTNOTE1 HEIGHT=2.2             /** FIRST GRAPH FOOTNOTE **/
      J=L ' SOURCE: MXG PERFORMANCE DATABASE'
      J=R 'SAS/GRAPH PLOT ';
FOOTNOTE2 HEIGHT=2.2             /** SECOND GRAPH FOOTNOTE**/
      J=L " MXG DATABASE FILE = " COLOR=RED "&MXGFILE DATA"
      J=R "&PLOTVAR1 ";

SYMBOL1 INTERPOL=JOIN C=CYAN L=1 W=3 VALUE=SQUARE H=4;
SYMBOL2 I=JOIN C=BLACK L=1 W=3 V=TRIANGLE H=4;
SYMBOL3 I=JOIN C=RED L=1 W=3 ;

LEGEND1 VALUE=(H=2.6 C=BLACK)     /**GRAPH LEGEND STATEMENT**/
      ACROSS=3 CFRAME=WH CBORDER=BLUE CSHADOW=GREY
      LABEL=(H=2.8 C=MAGENTA POSITION=(BOTTOM CENTER));
NOTE ' ' &TY71NOTE ;

PLOT (&PLOTVAR1 &PLOTVAR2 &PLOTVAR3)*TIME / OVERLAY
      AUTOVREF                      /** REF LINES VERTI AXIS **/
      AUTOHREF                      /** REF LINES HORIZ AXIS **/
      NAME="&PLOTVAR1"              /** NAME OF THE PLOT GIF **/
      ANNOTATE=MY_ANNOTATE          /** IDENTIFY ANNO DATASET **/
      HAXIS=AXIS2                   /** USE AXIS2 FOR HAXIS **/
      VAXIS=AXIS1                   /** USE AXIS1 FOR VAXIS **/
      VZERO                         /** START AXIS AT ZERO **/
      LEGEND=LEGEND1;               /** USE AND IDENTIFY LEGEND**/
RUN; QUIT;                          /** RUN AND CLOSE GPLOT **/
ODS HTML CLOSE;                     /** CLOSE HTML **/
ODS LISTING;                         /** LISTING RESET OPEN **/
RUN;
%MEND USER_TYPE71_PLOT;             /* END OF THE MACRO */

```